
Subject: Re: DHS Malawi - domestic violence among women 15-19

Posted by [Liz-DHS](#) on Tue, 29 Apr 2014 22:56:08 GMT

[View Forum Message](#) <> [Reply to Message](#)

Dear User,

Here is a response from one of our experts Kerry MacQuerrie (also please see her attachment)

****Tips for matching DV tables - Malawi 2010**

*Open file:

```
cd "C:\Users\kerry.macquarrie\Documents\HIV-DV - local\PAA2014\DV Forms\Data\" /*specify  
your directory location*/  
use "MWIR61FL.DTA", clear
```

*Weight data:

```
g wt=d005/1000000 /*This is the domestic violence weight, not the women's weight*/
```

*If adjusting for sampling design:

```
g strata=v023  
svyset [pw=wt], psu(v021) strata(strata) singleunit(center)
```

*Table 18.1 Experience of physical violence

/*"Ever experienced physical violence" draws from multiple variables:

Set 1: d105a-d105g & d105j. Women who answer 'yes' to these items are reporting her
current/most recent ///

husband/partner perpetuated this act of physical violence. They are then
asked how frequently in ///

the last 12 months. Take response codes 1-4 to any of these variables. ///

Set 2: d115y. Women who answer 'no' this variable are reporting that they have ever
been physically hurt ///

by someone other than their spouse since age 15. (It's reverse coded.) Take
response code 0. ///

Set 3: d118y. Women who answer 'no' to this variable are reporting that they have been
physically hurt ///

during a pregnancy. (It's also reverse coded.) Take response code 0. */

```
lab def yesno 0 "no" 1 "yes"
```

*create composite physical violence variable

```
g pever=.
```

```
replace pever=0 if v044==1
```

```
replace pever=1 if inrange(d105a,1,4) | inrange(d105b,1,4) | inrange(d105c,1,4) |  
inrange(d105d,1,4) | ///
```

```
inrange(d105e,1,4) | inrange(d105f,1,4) | inrange(d105g,1,4) | inrange(d105j,1,4)
```

```
replace pever=1 if d115y==0 | d118y==0
```

```
lab var pever "Ever experienced physical violence"
```

```
lab val pever yesno
```

*Check total

tab pvever [iw=wt] /*the results match the total reported in Table 18.1*/

*Restrict to 15-19 year olds

tab pvever if v013==1 [iw=wt]

*Table 18.4 Experience of sexual violence

/*"Ever experienced sexual violence" also draws from multiple variables:

Set 1: d105h and d105i captures sexual violence by current/most recent husband/partner ///

Take response codes 1-4.

Set 2: d123 captures if first sexual intercourse was forced. Take response code 2. ///

Set 3: d124 and d125 capture if anyone other than spouse forced sex in last 12 months or forced sex acts. ///

Take response code 1 to either. */

*create composite sexual violence variable

g svever=.

replace svever=0 if v044==1

replace svever=1 if inrange(d105h,1,4) | inrange(d105i,1,4)

replace svever=1 if d123==2

replace svever=1 if d124==1 | d125==1

lab var svever "Ever experienced sexual violence"

lab val svever yesno

*Check total

tab svever [iw=wt] /*matches total reported in Table 18.4*/

*Restrict to 15-19 year olds

tab svever if v013==1 [iw=wt]

*If you want to crosstab by background characteristics, check the DHS recode manual for a list of variables.

*Code would take the following form (I use age, v013, as an example):

tab v013 svever [iw=wt], row

File Attachments

1) [Malawi DV match.do](#), downloaded 910 times
